
Data handling and data transformations in SPSS
In this document I present an explanation of data handling techniques and data trans​formations in SPSS, along with a description of relevant syntax commands and some pull-down menu features. Exercises at the end of the document provide an opportunity to gain some experience with the procedures described. 

1.
A note on the use of syntax versus pull-down menus
When SPSS was first developed many years ago, menu-driven operating systems with graphic user surface such as windows had not been invented yet. Even PCs were not in existence. As a user, you would punch your data and SPSS command syntax on punch cards, usually one case or one command per card, run your stack of cards through a card reading machine that was con​nected to the main frame processor, and eventually the results of your computing job would be printed out. Later the punch cards were replaced by command lines in text editors, and you would type them and enter them into the main frame computer via a remote terminal. Then more powerful desktop computers were developed, and the computing would be done locally, but still the interaction with the computer was via text commands. Then Mac and Windows operating systems were developed, and along with their increasing popularity, statistics programmes like SPSS became (almost) fully controlled by graphical menus. 

So why not simply use the menus? Why bother learning syntax commands? There are several reasons. Some things simply cannot be done without syntax; other things may be done either way, but can be done more efficiently via syntax; and syntax files can be saved and re-used, whereas by using the menus, you have to start anew each time you run an analysis session. The basic structure of the SPSS programmes has not changed over the years. Whatever you do with the menus will create command syntax which is then read and processed. And it is still possible to do all your SPSS analyses by typing and running syn​tax commands, without using the menus at all. Most things can be done either way, although there are some features of SPSS that are not available through the menus and can only be done via typing syntax. We have seen examples of these, such as generating some outlier diagnostics or producing standard errors of beta coefficients in REGRESSION. Even some complete pro​cedures (e.g. MANOVA) can be accessed only via syntax. So the most sensible approach is to combine use of the menus and syntax commands in an efficient way. Especially data trans​form​ations can be done quite easily with a few commands that are easy to understand and apply.

2.
Creating datasets in SPSS
The first thing you need to do before you can analyse your data is create a dataset in SPSS-readable format. Whatever method you use to enter your data (see sections below), it is useful to outline the structure of your dataset in a coding scheme. For each variable, the coding scheme should contain information about its position in the dataset, some descriptive information, an SPSS variable name, information on the codes to be used for representing data values, and information on the handling of missing values, if any. An example is given in Table 1. Note that an SPSS variable name may consist of up to 8 characters (letters, numbers and / or some special characters) and must begin with a letter. (It may also begin with some special characters; e.g. $ for system variables created by SPSS).

You would normally list the variables in your coding scheme (and enter them in the dataset) in the same order as they appear in your questionnaire or research protocol.

Table 1:  Coding scheme example
	Columns

	Variable
	SPSS name

	Coding
	Missing

	1-3

	participant number
	PNUM
	001 to 125
	n/a

	4
	experimenter
	EXPE
	1=Alex, 2=Paula
	n/a

	5
	N of females in session
	NFEM
	(number)
	n/a

	6
	N of males in session
	NMAL
	(number)
	n/a

	7
	similarity condition
	 SIMC 
	1=similar, 2=dissimilar
	n/a

	8
	likability  rating
	LIKABLE
	1 to 7
	blank

	9
	"like to meet" rating
	MEET
	
	

	10
	similarity rating
	SIMILAR
	
	

	11
	crime condition
	CRIMC
	1=control, 2=rape,

3=assault
	n/a

	12
	assailant info given?
	ASS_INFO
	0=no 1=yes
	n/a

	13
	recall question 1
	rec1
	0=no response,

1=wrong,

2=part correct,

3=correct
	8 = recall questions not asked

9 =

unreadable

	14
	recall question 2
	rec2
	
	

	15
	recall question 3
	rec3
	
	

	16
	recall question 4
	rec4
	
	

	17
	sex of participant
	SEX
	1=m,  2=f
	blank

	18
	feeling state item 1
	F1
	1 to 7
	blank

	  ... to... 
	              ... to ...
	  ... to ...
	
	

	24
	feeling state item 7
	F7
	
	

	25-26
	age of participant
	AGE
	(in years)
	blank

	27
	suspicion?
	SUSP
	0=no, 1=yes
	8 = nothing written,

9=unreadable


2.1
Entering data

Two methods of data entry will be described. The first, using the SPSS data editor, is  relatively easy to follow and use, especially for beginners, but may be very time-consuming with large datasets. The second, entering data in a text file and later converting this for use in SPSS, is more complicated as it requires the knowledge of some syntax commands, but can be very efficient and economical with large datasets.

2.1.1
Using the SPSS data editor
Based on your coding scheme, you may enter the data directly into the SPSS data editor window in “data view” mode (an empty one pops up when you start SPSS). After double-clicking on a column header (originally desig​nated “var”), the display changes to “variable view” (the two “views” may be toggled by clicking on the “file card tabs” at the bottom left of the screen). Within “variable view”, enter your variable name, descriptive labels for the variable (e.g. text from the “variable” column of Table 1) and its values (see “coding” column in Table 1), as well as any information on missing values. You may also alter the column width and decimal places displayed (defaults are 8 characters wide and 2 decimal places). Variable type defaults to “numeric”, and it is useful to stick to simple numeric codes whenever possible.

Once the variable definitions have been entered, go back to “data view” and enter the data directly into the cells of the matrix. Each line of the data window represents one “case” (usually a participant), each column represents one variable. To get from one cell to the next or back, use the “tab”, “back tab” or arrow keys. When all the data have been entered, save the data file. SPSS will prompt you to do so when you attempt to leave the programme without saving. Of course you may enter your data in portions; new data can always be added to an existing data file. SPSS data files are also called system files and carry the default extension “.sav”.

2.1.2
Generating a data file in fixed ASCII (text only) format and converting it to an SPSS data file
Although the SPSS data entry facility is quite convenient, you should consider the alternative of creating a text file that contains your data, which can later be converted to an SPSS system file. This alternative may save a lot of time with large datasets. A text file (also called an ASCII file) has a rectangular format; as in the SPSS data window, cases are represented as lines and variables as columns or a range of columns.

A text file containing the dataset described in Table 1 may look like this:

001122155410303314255453170

002122155730333314253252180

003122155510333325544446180
      ..............

           etc.

      ..............

159211243220333323464355180

160211164120333117717117190

In this example, the first line tells us that participant number 001 was in a session where Alex was experimenter; that there were 2 females and three males in that session; that participant 001 was in the “similar” condition, gave a likability rating of 4 etc.

To create a text data file, you can use any word processing software, e.g. Word. Use a typefont with equal character spacing; that way it is easier to verify the correct line length and to spot errors.  (I have used Courier New above.) Make sure to hit “return” at the end of each line (this will later be read by SPSS as a line delimiter); if there are too many variables to fit in one line, use more than one line (but a fixed number of lines) per case. Save your dataset in text format. In Word, this is done by choosing “Save As” from the File menu and then changing the entry in the “Save as type” field at the bottom to “Text Only with Line Breaks”, then click on “Save”. A warning may pop up telling you that you may lose all of Word's wonderful formatting features when saving in this format -- simply click on “yes” and your text file will be saved.

To convert a text file into an active SPSS system file, use the DATA LIST command. Its general form is 

DATA LIST FILE=file RECORDS=n    

    /{1    } varname {col location [(format)]} [varname ...] 

     {rec #}         

   [/{2    } ...] [/ ...] 

     {rec #}                  .

[Note that SPSS syntax is not case-sensitive. For presentation purposes, within syntax examples I am using UPPERCASE for SPSS keywords and lowercase for variable names. Generally, syntax keywords may be abbreviated, usually down to the first three letters (except where this would be ambiguous, e.g. COMMENT vs. COMPUTE).]

After the keyword FILE=, enter the path and filename where you have saved your text file, including extension, enclosed in quotes (e.g. “a:\mydata\study1.txt”); after the keyword RECORDS=, enter the number of lines per case (in our above example, this would be 1). Then, after the slash, enter your variable names followed by their column location. Like all syntax commands, end the command with a dot.

If your text file contains more than one line per case, change the n parameter after RECORDS= accordingly, enter the variables in line 1 after the first slash, the variables in line 2 after the second slash etc. To facilitate this, you may enter the appropriate line number immediately after each slash, i.e. /1, /2, /3 etc.

In our example, the data list command would read:

DATA LIST FILE="a:\mydata\study1.txt"  REC=1

   / pnum 1-3  expe 4  nfem 5  nmal 6  simc 7  likable 8 

     meet 9  similar 10  crimc 11  ass_info 12 

     rec1 TO rec4 13-16  sex 17  age 18-19  f1 TO f7 20-26

     susp 27.

SPSS assumes by default that all your variables are coded by numeric values  (i.e. only numbers, no letter strings), and I would generally recommend to use only numeric codes. If, for some reason (though I cannot think of a good one), your text file contains alphanumeric values (e.g. “f” for female and “m” for male), this must be indicated in DATA LIST by an “a” in parentheses after the column location of the respective variable.

It is often convenient to code related variables with a common name stem followed by con​secutive numbers (as in rec1, rec2 etc.). Then you can define these variables in DATA LIST by using the keyword TO (as in f1 TO f7), followed by a range of columns that covers the width of all variables included. The TO convention can also be used in a referential sense in later commands. For example, once our dataset has been converted to an SPSS file, you may refer to the variables LIKABLE, MEET and SIMILAR as “likable TO similar”, e.g. in the variable list of a CORRELATION command. More examples for the use of the TO convention will be given below in the sections on data transformation.

Any blank field in a text dataset is automatically assigned a missing value, so the easiest way to code missing data is by entering spaces instead of numbers. Sometimes, however, you may want to express different types of missing data by assigning different values, as with the recall questions in the example: Here, "8" is used to indicate that for some reason the questions were not asked (perhaps due to a mistake by the experimenter), and "9" is used to denote a partici​pant's response as unreadable. These number-coded missing values must be actively defined by a MISSING VALUES command. Its general form is 

MISSING VALUES

 {varlist(value list)[[/]varlist ...]}

 {ALL(value)                         }  .

In our example, we would define all number-coded missing values with the following command:

MISSING VALUES  rec1 to rec4, susp (8, 9).
This should be entered after the DATA LIST command. If you use the same numeric values to indicate missing values for all variables in your dataset, you may conveniently type the keyword ALL  instead of a variable list containing all variables.

Finally, you should label your variables and values, at least those that are not self-explanatory, using the VARIABLE LABELS and VALUE LABELS commands. 

Their general forms are

VARIABLE LABELS varname 'label'   [/varname...]  .

and

VALUE LABELS varlist Value 'label' Value 'label'...

    [/varlist...]  .

In our example dataset, we might define labels as follows:

VAR LAB expe "experimenter" 

 / nfem "number of females in session" 

 / nmal "number of males in session" 

 / simc "similarity condition"          [... etc.] .

VAL LAB expe 1 "Alex" 2 "Paula" 

      / simc 1 "similar" 2 "dissimilar"

      / crimc 1 "control" 2 "rape" 3 "assault" 

      / susp 8 "nothing written" 9 "unreadable"  (... etc.)  .

Note that missing values can be assigned labels just as any other values.

Once the DATA LIST, MISSING VALUES, VARIABLE LABELS and VALUE LABELS have been run, your data are available in SPSS format and displayed in the data editor window.

2.2
Checking data entry errors
There are various methods of ensuring a relatively error-free representation of your data. At the data entry stage, you may decide to enter the most important variables twice for each case (e.g. relatively early and relatively late in a data line, representing each entry by different variable names. Later you would check the correlation of the two versions of each variable -- if they are all 1, there are probably no errors (unless the same error has been made twice in a given case, which is really unlikely).

To get an overall estimate of the accuracy of data entry, you may have a subset of cases entered by two independent coders and check their degree of correspondence.

An important tool for detecting and correcting data entry errors are plausibility checks using procedure FREQUENCIES. To check if all variable values are within their correct range, type  the following syntax:

FREQUENCIES ALL   /FORMAT  NOTABLE   /STATISTICS  MIN  MAX.
This produces a table containing the following summary information for all variables (note that keyword ALL is used here again): number of valid and missing cases (this is shown per default), minimum and maximum (generated by the keywords MIN and MAX in the STATISTICS subcommand). The /FORMAT NOTABLE subcommand (available only through syntax) suppresses the generation of frequency tables for each variable (which would be the default), as you are not interested in these at the moment.

What does the summary table show you? For each variable, you can check if their values are within the predefined range. E.g. for variable REC2 in our example, all values should be in the range 0 - 3, for variable LIKABLE, all values should be between 1 and 7. If, for example, you spotted a maximum value of 6 for REC2 or a minimum value of 0 for LIKABLE, there would be at least one coding error present for each variable. The next step would be to re-run procedure FREQUENCIES just for the variables that seem to be affected, this time generating only the default frequency tables:

FREQUENCIES rec2 likable.
This shows you exactly how many incorrectly coded cases there are on each variable. For example, you might find that for REC2 there is one case with a value of 6 and two cases with a coding of 5, whereas for LIKABLE, there are two cases with a value of 0. If your dataset is small, these cases can be easily located by scrolling down in the data window until you see the wrong codes, then find the original questionnaire based on the case number (always include case numbers and write them on the front pages of your questionnaires!) and correct the error. With larger datasets, it may be more convenient to have SPSS print a list of those cases that are affected by errors, using the LIST VARIABLES command. In our running example, the syntax required would be:

TEMPORARY.

SELECT IF (rec2 > 3 OR likable = 0).

LIST VARIABLES pnum.

This produces a list of the participant numbers (variable PNUM) just for those cases that are affected by errors (as specified in the argument of the SELECT IF command). The TEMPORARY command preceding SELECT IF causes the case selection to be applied only while executing the command immediately following SELECT IF. (NB: Without the TEMPORARY command, case selection would be permanent!)  

The plausibility check via FREQUENCIES is quite effective, though of course it does not help you to spot erroneous values that are within the allowed range of a variable (e.g. 6 instead of 3 for variable LIKABLE). 

3.
Data transformations
Data transformations are a vital step in almost all analyses. They are used to create new variables from existing ones, or to change existing variables. Examples: Combining ten items that form an attitude scale into an attitude score; reverse-scoring those items of a scale that have a negative wording; performing median splits; representing a categorial variable as a set of dummies; applying a logarithmic transformation to reduce skewness; forming a new variable that represents the number of occurrences of some qualitative response; etc.

I will emphasise the syntax commands COMPUTE, IF, COUNT and RECODE. These can be generated from the pull-down menu point Transform, which is visible in the data editor window (the COMPUTE and IF commands are both generated by the submenu Compute, the other two transformations by their own submenus). As usual, the menus may be easier to use for beginners and will do nicely for a few smaller transformations. As soon as you need to do many similar transformations, however, or as a large number of variables are involved in a given transformation, using syntax is to be recommended.

3.1
Common operators and functions
Before presenting the data transformation commands, we should have a look at the SPSS coding of some common arithmetic, logical and statistical operators and functions that can be used in these commands. I am listing here the most frequently used ones (for more on this, see SPSS syntax help window on COMPUTE).

Arithmetic Operators:
+
Addition

‑
Subtraction

*
Multiplication

/
Division

**
Exponentiation

Examples:
error**2
= the value of variable ERROR squared

f1 + f2

= the sum of the values of variables F1 and F2

Arithmetic Functions:
ABS
Absolute value
RND
Round

LG10
Base 10 log

LN
Natural log

SQRT
Square root

Examples:
abs(diff)
= the absolute value of variable DIFF

   [e.g., if DIFF = -3, then ABS(DIFF) = 3] 

rnd(attscale)
= the value of variable ATTSCALE rounded to the nearest integer

   [e.g., if ATTSCALE=4.32, then RND(ATTSCALE)=4]

Statistical Functions:
SUM[.n]

Sum of arguments

MEAN[.n]

Mean of arguments

Example:
mean (item1 to item4)
= the mean of ITEM1, ITEM2, ITEM3, ITEM4 

Missing‑Value Functions:
MISSING
True if missing

NMISS
Number of missing arguments

SYSMIS
True if system‑missing
NVALID
No. of valid arguments

Arithmetic Comparators:
=    (or  EQ)

“is equal to”

>    (or GT)
“is greater than”

<>  (or NE)

“is not equal to”

<=  (or LE)
“is less than or equal to”

<    (or LT)

“is less than”

>=  (or GE)
“is greater than or equal to”

Logical Operators:

AND (or &)

= logical “and”

OR


= logical “or”

NOT(expression)
= “expression is not true” (specified within a condition argument)

(For examples of how to use missing-value functions, comparators and logical operators see section on IF command below.)

3.2
Data transformation commands
Below I will describe the most common data transformation commands. You should note that any data transformations you run will not take effect until SPSS encounters the next analysis command. So don’t be confused if you have run, e.g., a series of compute commands that ought to have generated new variables, and all you see in the data editor is empty columns. If you want your transformations to take effect immediately, simply add another syntax line that reads

EXECUTE.

after your last data transformation line. This causes SPSS to write any pending transformation outcomes to the active data file. (The same effect is achieved by selecting from the menus: Transform – Run Pending Transforms.)

3.2.1
COMPUTE
The COMPUTE command is used to define a new variable or change an existing one by computing new values. Its general form is

COMPUTE  target variable = expression.
“Target variable” can be an existing variable (if you want to change this variable's values) or a new variable, which is defined and added to the data file by running the COMPUTE command.

“Expression” is an arithmetic expression that may contain variable names, numbers, operators and functions. The numerical value of  “expression” is assigned to the “target variable” for each case.

Examples:
COMPUTE  item5 = 8 - item5. 

Changes the values of the existing variable ITEM5. If ITEM5 is measured on a 1‑7 scale, the above command can be used for reverse-scoring the item (value 1 becomes 7, 2 becomes 6 etc.)

COMPUTE  item5r = 8 - item5. 

Same as above, but this time a new variable ITEM5R is created, while ITEM5 is preserved in its original scoring.

COMPUTE  attscale = mean(att1, att2, att3).

Computes new variable ATTSCALE as the mean of items ATT1, ATT2 and ATT3.

COMPUTE  memoryx = mean.30(trial1 to trial50).

Computes new variable MEMORYX as the mean of items TRIAL1 to TRIAL50 if at least 30 of those items are not missing. (An integer number after a dot following a statistical function keyword specifies the minimum number of valid arguments for the variable to be computed.) Cases that have less than 30 valid values in TRIAL 1 to TRIAL50 are assigned a missing value on MEMORYX.

COMPUTE  logrt1 = lg10(rtime1).
Assigns the logarithmic transformation of variable RTIME1 to new variable LOGRT1.

3.2.2
IF

The IF command computes data transformations just like the COMPUTE command, but allows you to specify a condition which must be true for the data transformation to be performed. Its general form is

IF [(]logical expression[)] target variable=expression
The logical expression contains at least one variable name; it may further contain numbers, arithmetic and logical operators. A typical example would be performing a median split, i.e. creating a dichotomous variable and assigning values to “low” and “high” levels of the new variable based on whether the values on an existing continuous variable are below or above the median.Assume that you have determined the median of variable PREJDICE to be 4.37 (this can be done via procedure FREQUENCIES), and you want to create a new variable PREJSPLT that represents low versus high levels of PREJDICE. To do so, you could specify

IF  (prejdice <  4.37)  prejsplt = 0.

IF  (prejdice >= 4.37)  prejsplt = 1.

VAR LABELS   prejsplt  "dichotomised prejudice measure" .

VAL LABELS   prejsplt  0 "low"  1 "high" .
Another example: You want to create a new variable RESP that represents whether a participant has res​ponded to an item about their sexual preference (variable SEXPREF) or not. You specify

COMP  resp = 1.

IF (MISSING(sexpref))  resp = 0.

Note that the COMPUTE command generates the new variable RESP and sets its value to 1 for all cases. The subsequent IF command overwrites this value with 0 for those cases that have a missing value on variable SEXPREF.

The logical expression in an if command may specify complex conditions containing logical operators, as in the following example, referring to the dataset described in Table 1:

IF ((sex=1 and crimc=3) or (sex=2 and crimc=2)) match = 1. 

IF ((sex=1 and crimc=2) or (sex=2 and crimc=3)) match = 0.

These commands create a variable MATCH that indicates whether sex of participant (SEX) and sex of the victim in a crime vignette that participants read (CRIMC) are the same (1) or different (0). Note that participants in the control condition (CRIMC=1), who did not read about a crime, are not assigned values and thus will receive system-missing values on the new variable MATCH.

3.2.3
COUNT
The COUNT command counts, for each case, the occurrence of a given value or range of values in a variable or list of variables and assigns the result as a value to a target variable. Its general form is

COUNT target variable = varlist(value list)   
[In the value list, you may use the keywords LOWEST (LO), HIGHEST (HI), THRU, MISSING and SYSMIS.]

Example: You have asked participants to describe up to five behaviours they engage in when under stress. The responses are later coded as variables BEH1 to BEH5, each with values 1 (problem-focused coping), 2 (emotion-focused coping), 3 (avoidance), 4 (other). You want to create a new variable NUMCOPE that represents the number of coping behaviours (categories 1 and 2) listed. You specify 

COUNT  numcope = beh1 TO beh5 (1, 2) .
In another study, the content and valence of  up to 10 thoughts that participants listed has been coded as 10 two-digit variables THOU1 to THOU10, whose first digit represents content (1=self-related, 2=other-related) and whose second digit represents valence (1=positive, 2=negative, 3=neutral). The following IF commands generate new variables that represent the number of self-related thoughts and the number of negative thoughts, respectively:

COU  numself =  thou1 TO thou10 (11 THRU 13).

COU  numneg =  thou1 TO thou10  (12, 22).

The first of these commands could also have been written

COU  numself =  thou1 TO thou10 (LOWEST THRU 13).

as 11 is the lowest possible value of THOU1 to THOU10.

3.2.4
RECODE
The RECODE command is used to change the values of a given variable. Its most relevant syntax is 

RECODE varlist (old value list= new value)    

               [(old value list=new value)...

       [INTO varlist] .

[Keywords you may use in the old value list are LO, LOWEST, HI, HIGHEST, THRU, MISSING, SYSMIS, and ELSE. Keywords you may use instead of numeric new values are COPY and SYSMIS.] 

Assume you want to reverse-score each of three negatively worded attitude items (ATT3, ATT7 and ATT9), measured on a scale from 1 to 5. You could specify

RECODE att3 att7 att9 (1=5)(2=4)(4=2)(5=1).
If you want to preserve the original variables and add reverse-coded variables with new names, you may specify

RECODE att3 att7 att9 (1=5)(2=4)(3=3)(4=2)(5=1) 

       INTO att3r att7r att9r.
This command creates the new variables ATT3R, ATT7R and ATT9R that are the reverse-coded mirror-images of ATT3, ATT7 and ATT9. Note that when using the INTO keyword, all old values must be explicitly included, hence the specification “(3=3)”. If you omitted this as in the first command (where it may be omitted), the new variables would be assigned missing values for those cases where the old variables were coded 3. 

As another example, consider that you have a qualitative variable CAT representing 7 categories, coded 1 through 7; you want to combine the first two categories into one, leave the next 3 as they are, and assign the remaining two as missing values on a new variable CATNEW. Your RECODE command would read

RECODE cat (1,2=1)(3 THRU 5=COPY)(ELSE=SYSMIS) INTO catnew.
This demonstrates the proper usage of keywords THRU, COPY, ELSE and SYSMIS. The keyword COPY tells SPSS that the new values are the same as the old values; the keyword ELSE stands for all the old values that have not been specified yet; SYSMIS assigns a system-missing value. 

3.3
DO REPEAT structure
Often you have to perform a series of very similar data transformations. Going back to one of the examples in the section on COMPUTE, assume that you want to log-transform response times from 100 trials, called RTIME1 to RTIME100. Of course you could type the first COMPUTE command to transform RTIME1, then copy and paste this command 99 times and overwrite the numbers in each command line -- this would be faster than mouse-clicking your way through the Transform - Compute pull-down menu 100 times. An even quicker way is by using what is called a DO REPEAT - END REPEAT structure (only available in syntax), which does the copying and pasting for you. All you need to do is define “stand-in variables” for a range of existing and / or target variables. The data transformation is then specified only once for the stand-in variables. The general form of this structure is 

DO REPEAT stand‑in var=varlist [/stand‑in var=...]

transformation commands

END REPEAT [PRINT] .

In our example, we would specify

DO REPEAT  y = logrt1 to logrt100  / x = rtime1 to rtime100 

COMPUTE    y = lg10 (x)

END REPEAT PRINT.

Note that stand-in variable X represents the 100 existing raw response times and stand-in variable Y represents 100 log-transformed target variables. The transformation is specified once in the COMPUTE command between the DO REPEAT and END REPEAT lines. This structure generates a sequence of 100 commands. The PRINT subcommand lists the sequence of resulting commands in the log as follows:

COMPUTE  logrt1 = lg10(rtime1).

COMPUTE  logrt2 = lg10(rtime2).

            :

            :

COMPUTE  logrt99 = lg10(rtime99).

COMPUTE  logrt100 = lg10(rtime100).

4.
Further useful procedures for data handling
4.1
Combining existing SPSS datasets
Combining datasets via the pull-down menus can be quite convenient, as long as only two or three datasets are being combined. For combining larger numbers of datasets, syntax commands provide flexible alternatives. Both methods will be described.

4.1.1
Datasets containing identical variables and different cases
This is done by clicking on Data - Merge Files - Add Cases, or by using the ADD FILES syntax command. Examples: The same variables have been assessed in two independent replication studies; data from each study are available in separate SPSS files. Or you and a friend have shared the job of data entry and have each entered half of the cases  of one study into different files.

When using the pull-down menus, one of the files to be combined must be open. Then, from within the data window, select  Data - Merge Files - Add Cases. In the AddCases: File Name window that pops up, select or type the name of the file to be added and click on Open. In the Add Cases from... window that pops up, SPSS shows you the variables that are common to both files, which have already been incuded in the new working data file (right half of the window), and those variables that appear in only one of the two files (“unpaired variables”; left half of the window). Variables may only seem to be “unpaired” if different names have been assigned to the same variable in each of the two files. This can be remedied by selecting a variable in the left-hand list, clicking on Rename and renaming the variable so the two names become identical. You can move unpaired variables to the right half of the window to have them included, or move paired variables to the left to exclude them from the working data file. You can also have the file from which the new cases come indicated as a new variable, by ticking “Indicate case source as variable” and entering a variable name. 

Once you are happy with the variable names in the right-hand list, click on OK, and the combined file will become your active data file. Save that file under a new name. To combine more than two files, the steps just described must be repeated for each file to be added. Or do it all in one go using syntax.

The most relevant syntax elements are

ADD FILES FILE=file[/RENAME=(old varlist=new varlist)...]

                   [/IN=varname]

        [/FILE=...][/RENAME...][/IN...]

        [/KEEP={ALL    }] [/DROP=varlist]

               {varlist}

At least two FILE subcommands must be specified. Files named in a FILE subcommand can be  any SPSS system file, including the currently active file (if any). List path and filename between quotes, or use an asterisk (*) to refer to the currently active file.

Example 1:

ADD FILES FILE=* /FILE="a:\mydata\second_file.sav". 

This command combines the currently active file with the file named c:\mydata\second_file.sav by appending all cases of the latter to the end of the former. This is done even if the two files do not contain the same variables. On variables that are defined in one file but not the other, missing values will be assigned to cases from the other file.

The /RENAME subcommand can be used to rename variables in the file named in the preceding FILE subcommand. This becomes necessary if identical variables carry different names. Example: The variable “attitudes toward women” is named ATTWOM in the currently active file, and is named ATW in c:\mydata\second_file.sav. To preserve the variable information from both files in a variable named ATW in the resulting file, you would specify:

ADD FILES FILE=* /RENAME= (attwom = atw)

         /FILE="a:\mydata\second_file.sav". 

The /IN subcommand allows you to create a new variable that will have a value of 1 for all cases that come from the file named in the preceding FILE subcommand, and a value of 0 for cases from any other file. For example, to combine RPS (research participation scheme) datasets from three consecutive years and preserve the year information in the final dataset, one might specify:

ADD FILES FILE= "c:\rpsdata\rps1998" /IN= year1998

         /FILE= "c:\rpsdata\rps1999" /IN= year1999

         /FILE= "c:\rpsdata\rps2000" /IN= year2000. 

The /KEEP or /DROP subcommands can be used to specify a subset of variables to be included or excluded. Example: You have two datasets each containing the variables HEALTH01 to HEALTH10, plus a host of other variables. You want to carry out further analyses with the HEALTH variables from both datasets combined, but are not interested in any of the other variables. Thus, at the end of your ADD FILES command, you would specify:

              /KEEP health01 to health10

The default setting is that SPSS keeps all variables appearing in at least one file.

The /MAP subcommand produces a listing of the variables in the new file and their locations in the log.

4.1.2
Datasets containing identical cases and different variables
The two datasets must contain information on the same cases in exactly the same order. If the cases are not in the same order initially, you must sort them. To do so, it is useful to have a variable representing case number appear in both datasets; this is used as the sorting variable. Open each file, run a SORT CASES command, and save the resulting file. Assuming that your case number variable is called CNUM, this command would simply read

SORT CASES BY cnum. 
Once this has been done for all the files to be combined, you can proceed by opening one file, clicking on Data - Merge Files - Add Variables, or by using the MATCH FILES syntax command. In the menu-based variant, things are very similar to the procedure for adding cases described above. An Add Variables: File Name window pops up, then you select or type the name of the file to be added and click on Open. In the Add Variables from... window that pops up, SPSS shows you the variables of the old and new file, which now jointly make up the new working data file in the right half of the window, and any overlapping variables (usually just your sorting variable) in the left half of the window. At that point, you would normally click on OK, and finally save the combined file under a new name. To combine more than two files, the steps just described must be repeated for each file to be added. Or, use the MATCH FILES syntax command.

The most relevant syntax elements are

MATCH FILES {FILE  }=file [/RENAME=(old varlist=new varlist)...]

          [/{FILE  }=...]

 [/MAP]

 [/KEEP={ALL    }] [/DROP=varlist]

        {varlist}

The meaning of the subcommands shown above is the same as in ADD FILES.

4.2
Writing an ASCII file from an existing SPSS file
It is sometimes useful to convert an existing SPSS data file to an ASCII (text only) file. This procedure, which is the reverse of what is described in section 2.1.2 above, can help you to redefine the units of analysis in your data. Assume that you have gathered data from married couples and entered each person’s responses as a case in an SPSS file in order to analyse the data at an individual level. Later you decide that it would be useful to analyse the data at the level of couples rather than individuals. This is problematic with the setup of your SPSS data file, because you now want to treat each couple as a “case”, and the variables for one case should all be in one line (not two lines) of the data editor. A relatively easy way to achieve this is to follow 3 steps:

(a) sort your individual cases so that husband and wife always appear in consecutive lines, 

(b) convert your data set to an ASCII file using the WRITE command,

(c) re-define this ASCII file using a DATA LIST command that now specifies the data from one couple as one case.

This is illustrated by the following mini-file, consisting of 4 individuals (2 couples):

The five variables in this file represent a personal participant number, a number for each couple, sex of participant, a happiness rating and a satisfaction rating. Note that the file is not sorted by couple. To sort it in a way that husband and wife appear in consecutive lines and always in the same within-couple order, we use the commands

SORT CASES BY coupnum sex .

EXECUTE.

This produces the following arrangement of data:

The data are now sorted by couple (variable COUPNUM) and within couple by SEX.

To produce an ASCII file, we use the command WRITE. Its general form is very similar to that of DATA LIST:

WRITE  [OUTFILE=file] [RECORDS={1}]

                               {n}

       /{1    } varlist [{column location [(format)]}]

        {rec #}          

       /{2    }...] 

        {rec #...}   .
In our example, the command might read

WRITE OUT = "a:\mydata\couples.dat"  RECORDS = 1

    / persnum 1  coupnum 2  sex 3  happy 4  satisfd 5.

EXE.

This produces the following ASCII file:

11156

21265

32136

42225
This file can now be redefined to represent couples as units of observation (or “cases”):

DATA LIST  FIL= "a:\mydata\couples.dat"  RECORDS = 2

   /1  coupnum 2  hus_hap 4  hus_sat 5

   /2  wif_hap 4  wif_sat 5.

EXE.
The resulting SPSS system file looks like this:

It contains two cases (= couples) and features the happiness and satisfaction ratings of husband and wife, respectively. Note that these variables had to be renamed to unambiguously relate them to each person in the couple. The SEX and PERSNUM variables were not coded at all, as it is now clear from the variable names which rating comes from which person within a couple, and couples are identified by the COUPNUM variable.

5.
Exercises

Exercise 1
Access the questionnaire form v:\courses\sp801\questionnaire.doc. This is taken from a practical study on sex differences in responses to infidelity. Generate a coding scheme for responses to this questionnaire, similar to the one displayed in Table 1 (see section 2 above). 

Exercise 2

(a) Generate an SPSS system file from the attached ASCII file named “exercise.txt”, which contains the data described in the coding scheme of Table 1. Assign variable labels and value labels (where appropriate); also define missing values where necessary (see section 2.1.2 above – commands DATA LIST, MISSING VALUES, VARIABLE LABELS and VALUE LABELS).

(b) There are some coding errors in this file. Try to find them and report which cases have erroneous values on which variables (see section 2.2 above).

(c) As you do not have access to the original raw data, you cannot correct the errors. All you can do in this case is to turn the erroneous values into missing values. Do so by overwriting them with blanks in the SPSS data editor window. Then save the resulting file on a floppy disc for further use. Save your syntax file as well.

Exercise 3
(a) Open the file created in exercise 2 (unless it is still open). Generate a new variable MEANRAT that represents the mean of the likability, “like to meet” and similarity ratings (see section 3.2.1 above). Define a variable label for your new variable (see section 2.1.2 above).

(b) What are the mean, standard deviation and number of valid cases for variable MEANRAT? How did you generate this information?

(c) Generate a new variable NUMCORR that represents the number of times a participant gave correct or partially correct responses to the four recall questions (see section 3.2.3 above). Label that variable. 

(d) Is the number of correct or partially correct responses related to participants’ age? How did you test this?

(e) Generate a new variable OLDYOUNG that represents a median split on participants’ age (see section 3.2.2 above).

(f) Use the RECODE command to generate a new CONDITN variable in which the rape and assault conditions coded in CRIMC are combined to form one condition, while the control condition is preserved as it is (see section 3.2.4 above).

(g) Save your SPSS data file. Then have SPSS write to your floppy disc a new ASCII file that only contains the variables PNUM, CONDITN, MEANRAT and NUMCORR (see section 4.2 above).

(h) Save and print the syntax that you have generated to do Exercise 3.

Appendix

Joining files in SPSS

1.
Open one of the files to be combined. 

2.
Select  Data - Merge Files - Add Cases. 

3.
In the “AddCases: File Name” window that pops up, select or type the name of one file to be added and click on Open. 

4.
Click on OK, and the combined file will become your active data file. 

5.
For each additional file, repeat the steps just described.

6.
Save the final dataset under a new name.







